

Monadically Making Modules

3-for-1 Monadic Notation: Do-it-yourself module types

MUSA AL-HASSY, JACQUES CARETTE, WOLFRAM KAHL

Can parameterised records and algebraic datatypes be derived from one pragmatic declaration?

Record types give a universe of discourse, parameterised record types fix parts of that universe ahead of time, and algebraic datatypes give us first-class syntax, whence evaluators and optimisers.

The answer is in the affirmative. Besides a practical shared declaration interface, which is extensible in the language, we also find that common data structures correspond to simple theories.

ACM Reference Format:

Musa Al-hassy, Jacques Carette, Wolfram Kahl. 2020. Monadically Making Modules 3-for-1 Monadic Notation: Do-it-yourself module types. 1, 1 (March 2020), 15 pages. <https://doi.org/10.1145/nnnnnnnn.nnnnnnnn>

1 INTRODUCTION

We routinely write algebraic datatypes to provide a first-class syntax for record values. We work with semantic values, but need syntax to provide serialisation and introspection capabilities. A concept is thus rendered twice, once at the semantic level using records and again at the syntactic level using algebraic datatypes. Even worse, there is usually a need to expose fields of a record at the type level and so yet another variation of the same concept needs to be written. Our idea is to unify the various type declarations into one —using monadic do-notation and in-language meta-programming combinators to then extract possibly parameterised records and algebraic data types.

For example, there are two ways to implement the type of graphs in the dependently-typed language Agda [4, 11]: Having the vertices be a parameter or having them be a field of the record. Then there is also the syntax for graph vertex relationships.

```
record Graph0 : Set1 where
  constructor ⟨_, _⟩0
```

Author’s address: Musa Al-hassy, Jacques Carette, Wolfram Kahl.

Permission to make digital or hard copies of all or part of this work for personal or classroom use is granted without fee provided that copies are not made or distributed for profit or commercial advantage and that copies bear this notice and the full citation on the first page. Copyrights for components of this work owned by others than ACM must be honored. Abstracting with credit is permitted. To copy otherwise, or republish, to post on servers or to redistribute to lists, requires prior specific permission and/or a fee. Request permissions from permissions@acm.org.

© 2020 Association for Computing Machinery.

Manuscript submitted to ACM

Manuscript submitted to ACM

```

field
  Vertex : Set
  Edges : Vertex → Vertex → Set

record Graph1 (Vertex : Set) : Set1 where
  constructor ⟨_⟩1
  field
    Edges : Vertex → Vertex → Set

data Graph (Vertex : Set) : Set where
  ⟨_,_⟩s : Vertex → Vertex → Graph Vertex

```

To illustrate the difference of the first two, consider the function `comap`, which relabels the vertices of a graph, using a function `f` to transform vertices:

```

comap0 : {A B : Set}
  → (f : A → B)
  → (Σ G : Graph0 • Vertex G ≡ B)
  → (Σ H : Graph0 • Vertex H ≡ A)
comap0 {A} f (G , refl) = ⟨ A , (λ x y → Edges G (f x) (f y)) ⟩0 , refl

comap1 : {A B : Set}
  → (f : A → B)
  → Graph1 B
  → Graph1 A
comap1 f ⟨ edges ⟩1 = ⟨ (λ x y → edges (f x) (f y)) ⟩1

```

In `comap0`, the input graph `G` and the output graph `H` have their vertex sets constrained to match the type of the relabelling function `f`. Without the constraints, we could not even write the function for `Graph0`. With such an importance, it is surprising to see that the occurrences of the constraint proofs are uninteresting `refl`-equality proofs. In contrast, `comap1` does not carry any excess baggage at the type level nor at the implementation level.

We will show an automatic technique for obtaining the above three definitions of graphs from a single declaration using similar notation. Our contributions are to show:

- (1) Languages with sufficiently powerful type systems and meta-programming can conflate record and termtype declarations into one practical interface. We identify the problem and the subtleties in shifting between representations in Section 2.
- (2) Parameterised records can be obtained on-demand from non-parameterised records (Section 3).

3-for-1 Monadic Notation: Do-it-yourself module types

3

- As with Graph_0 , the traditional [7] approach to unbundling a record requires the use of transport along propositional equalities, with trivial refl -exivity proofs. In Section 3, we develop a combinator, $_:\text{waist}_$, which removes the boilerplate necessary at the type specialisation location as well as at the instance declaration location.

(3) Programming with fixed-points of unary type constructors can be made as simple as programming with `termtypes` (Section 4).

- Astonishingly, we mechanically regain ubiquitous data structures such as `N`, `Maybe`, `List` as the `termtypes` of simple pointed and monoidal theories.

As an application, in Section 5 we show that the resulting setup applies as a semantics for a declarative pre-processing tool that accomplishes the above tasks.

2 THE PROBLEMS

There are a number of problems, with the number of parameters being exposed being the pivotal concern. To exemplify the distinctions at the type level as more parameters are exposed, consider the following approaches to formalising a dynamical system —a collection of states, a designated start state, and a transition function.

```
record DynamicSystem0 : Set1 where
  field
    States : Set
    start  : States
    next   : States → States
```

```
record DynamicSystem1 (States : Set) : Set where
  field
    start : States
    next  : States → States
```

```
record DynamicSystem2 (States : Set) (start : States) : Set where
  field
    next : States → States
```

Each `DynamicSystemi` is a type constructor of i -many arguments; but it is the types of these constructors that provide insight into the sort of data they contain:

Type	Kind
<code>DynamicSystem₀</code>	<code>Set₁</code>
<code>DynamicSystem₁</code>	$\prod X : \text{Set} \bullet \text{Set}$
<code>DynamicSystem₂</code>	$\prod X : \text{Set} \bullet \prod x : X \bullet \text{Set}$

We shall refer to the concern of moving from a record to a parameterised record as **the unbundling problem** [5]. For example, moving from the *type* Set_1 to the *function type* $\Pi X : \text{Set} \bullet \text{Set}$ gets us from DynamicSystem_0 to something resembling DynamicSystem_1 , which we arrive at if we can obtain a *type constructor* $\lambda X : \text{Set} \bullet \dots$. We shall refer to the latter change as *reification* since the result is more concrete, it can be applied; it will be denoted by $\Pi \rightarrow \lambda$.

Of-course, there is also the need for descriptions of values, which leads to the following termtypes. We shall refer to the shift from record types to algebraic data types as **the termtype problem**.

data $\text{DTerms}_0 : \text{Set}$ **where**

start : DTerms_0

next : $\text{DTerms}_0 \rightarrow \text{DTerms}_0$

data $\text{DTerms}_1 (\text{States} : \text{Set}) : \text{Set}$ **where**

start : $\text{States} \rightarrow \text{DTerms}_1 \text{ States}$

next : $\text{DTerms}_1 \text{ States} \rightarrow \text{DTerms}_1 \text{ States}$

data $\text{DTerms}_2 (\text{States} : \text{Set}) (\text{start} : \text{States}) : \text{Set}$ **where**

next : $\text{DTerms}_2 \text{ States start} \rightarrow \text{DTerms}_2 \text{ States start}$

Table 1. Contexts embody all kinds of grouping mechanisms

Concept	Concrete Syntax	Description
Context	$\text{do } S \leftarrow \text{Set}; s \leftarrow S; n \leftarrow (S \rightarrow S); \text{End}$	“name-type pairs”
Record Type	$\Sigma S : \text{Set} \bullet \Sigma s : S \bullet \Sigma n : S \rightarrow S \bullet 1$	“bundled-up data”
Function Type	$\Pi S \bullet \Sigma s : S \bullet \Sigma n : S \rightarrow S \bullet 1$	“a type of functions”
Type constructor	$\lambda S \bullet \Sigma s : S \bullet \Sigma n : S \rightarrow S \bullet 1$	“a function on types”
Algebraic datatype	$\text{data } D : \text{Set} \text{ where } s : D; n : D \rightarrow D$	“a descriptive syntax”

Our aim is to obtain all of these notions —of ways to group data together— from a single user-friendly context declaration, using monadic notation.

3 MONADIC NOTATION

There is little use in an idea that is difficult to use in practice. As such, we conflate records and termtypes by starting with an ideal syntax they would share, then derive the necessary artefacts that permit it. Our choice of syntax is monadic do-notation [10?]:

DynamicSystem : Context ℓ_1

DynamicSystem = **do** $X \leftarrow \text{Set}$

$z \leftarrow X$

$s \leftarrow (X \rightarrow X)$

End

3-for-1 Monadic Notation: Do-it-yourself module types

5

Here `Context`, `End`, and the underlying monadic bind operator are unknown. Since we want to be able to *expose* a number of fields at will, we may take `Context` to be types indexed by a number denoting exposure. Moreover, since records are a product type, we expect there to be a recursive definition whose base case will be the essential identity of products, the unit type `1`.

Exposure	Elaboration
0	$\Sigma X : \text{Set} \bullet \Sigma z : X \bullet \Sigma s : (X \rightarrow X) \bullet 1$
1	$\Pi X : \text{Set} \bullet \Sigma z : X \bullet \Sigma s : (X \rightarrow X) \bullet 1$
2	$\Pi X : \text{Set} \bullet \Pi z : X \bullet \Sigma s : (X \rightarrow X) \bullet 1$
3	$\Pi X : \text{Set} \bullet \Pi z : X \bullet \Pi s : (X \rightarrow X) \bullet 1$

With these elaborations of `DynamicSystem` to guide the way, we resolve two of our unknowns.

```
{- “Contexts” are exposure-indexed types -}
```

```
Context = λ ℓ → ℕ → Set ℓ
```

```
{- Every type is a context -}
```

```
‘_ : ∀ {ℓ} → Set ℓ → Context ℓ
```

```
‘S = λ _ → S
```

```
{- The “empty context” is the unit type -}
```

```
End : ∀ {ℓ} → Context ℓ
```

```
End = ‘ 1
```

It remains to identify the definition of the underlying bind operation `>>=`. Classically, for a type constructor `m`, bind is typed $\forall \{X \ Y : \text{Set}\} \rightarrow m \ X \rightarrow (X \rightarrow m \ Y) \rightarrow m \ Y$. It allows one to “extract an `X`-value for later use” in the `m Y` context. Since our `m = Context` is from levels to types, we need to slightly alter bind’s typing.

```
_>>=_ : ∀ {a b}
```

```
→ (Γ : Context a)
```

```
→ (∀ {n} → Γ n → Context b)
```

```
→ Context (a ⊔ b)
```

```
(Γ >>= f) ℕ.zero = Σ γ : Γ 0 • f γ 0
```

```
(Γ >>= f) (suc n) = (γ : Γ n) → f γ n
```

The definition here accounts for the current exposure index: If zero, we have *record types*, otherwise *function types*. Using this definition, the above dynamical system context would need to be expressed using the lifting quote operation.

```
‘Set >>= λ X → ‘ X >>= λ z → ‘ (X → X) >>= End
```

```
{- or -}
```

```

do X ← ‘ Set
  z ← ‘ X
  s ← ‘ (X → X)
End

```

Interestingly [3, 8], use of do-notation in preference to bind, $\gg=$, was suggested by John Launchbury in 1993 and was first implemented by Mark Jones in Gofer. Anyhow, with our goal of practicality in mind, we shall “build the lifting quote into the definition” of bind:

```

_>>=_ : ∀ {a b}
  → (Γ : Set a) -- Main difference
  → (Γ → Context b)
  → Context (a ⊔ b)
(Γ >>= f) N.zero = Σ γ : Γ • f γ 0
(Γ >>= f) (suc n) = (γ : Γ) → f γ n

```

With this definition, the above declaration `DynamicSystem` typechecks. However, `DynamicSystem i` $\not\equiv$ `DynamicSystemi`, instead `DynamicSystem i` are “factories”: Given i -many arguments, a product value is formed. What if we want to *instantiate* some of the factory arguments ahead of time?

```

N0 : DynamicSystem 0 {- ≈ Σ X : Set • Σ z : X • Σ s : (X → X) • 1 -}
N0 = N , 0 , suc , tt

```

```

N1 : DynamicSystem 1 {- ≈ Π X : Set • Σ z : X • Σ s : (X → X) • 1 -}
N1 = λ X → ??? {- Impossible to complete if X is empty! -}

```

```

{- “Instantiating” X to be N in “DynamicSystem 1” -}
N1' : let X = N in Σ z : X • Σ s : (X → X) • 1
N1' = 0 , suc , tt

```

It seems what we need is method, say $\Pi \rightarrow \lambda$, that takes a Π -type and transforms it into a λ -expression. One could use a universe, an algebraic type of codes denoting types, to define $\Pi \rightarrow \lambda$. However, one can no longer then easily use existing types since they are not formed from the universe’s constructors, thereby resulting in duplication of existing types via the universe encoding. This is not practical nor pragmatic.

As such, we are left with pattern matching on the language’s type formation primitives as the only reasonable approach. The method $\Pi \rightarrow \lambda$ is thus a macro that acts on the syntactic term representations of types.

```

Π → λ (Π a : A • τ) = (λ a : A • τ)
{- One then extends this homomorphically over all possible term formers. -}

```

That is, we walk along the term tree replacing occurrences of Π with λ . For example,

```

 $\Pi \rightarrow \lambda$  ( $\Pi \rightarrow \lambda$  (DynamicSystem 2))
≡{- Definition of DynamicSystem at exposure level 2 -}
 $\Pi \rightarrow \lambda$  ( $\Pi \rightarrow \lambda$  ( $\Pi X : \mathbf{Set} \bullet \Pi s : X \bullet \Sigma n : X \rightarrow X \bullet 1$ ))
≡{- Definition of  $\Pi \rightarrow \lambda$  -}
 $\Pi \rightarrow \lambda$  ( $\lambda X : \mathbf{Set} \bullet \Pi s : X \bullet \Sigma n : X \rightarrow X \bullet 1$ )
≡{- Homomorphism of  $\Pi \rightarrow \lambda$  -}
 $\lambda X : \mathbf{Set} \bullet \Pi \rightarrow \lambda$  ( $\Pi s : X \bullet \Sigma n : X \rightarrow X \bullet 1$ )
≡{- Definition of  $\Pi \rightarrow \lambda$  -}
≡  $\lambda X : \mathbf{Set} \bullet \lambda s : X \bullet \Sigma n : X \rightarrow X \bullet 1$ 

```

For practicality, `_:waist_` is a macro acting on contexts that repeats $\Pi \rightarrow \lambda$ a number of times in order to lift a number of field components to the parameter level.

```

 $\tau$  :waist n      =  $\Pi \rightarrow \lambda^n$  n ( $\tau$  n)

```

```

 $\Pi \rightarrow \lambda^n$  0       $\tau$  =  $\tau$ 
 $\Pi \rightarrow \lambda^n$  (n + 1)  $\tau$  =  $\Pi \rightarrow \lambda^n$  n ( $\Pi \rightarrow \lambda$   $\tau$ )

```

We can now “fix arguments ahead of time”. Before such demonstration, we need to be mindful of our practicality goals: One declares a grouping mechanism with `do . . . End`, which in turn has its instance values constructed with `< . . . >`.

```
-- Expressions of the form “... , tt” may now be written “< ... >”
```

```
infixr 5 < _>
```

```
<> :  $\forall \{\ell\} \rightarrow 1 \{\ell\}$ 
```

```
<> = tt
```

```
< :  $\forall \{\ell\} \{S : \mathbf{Set} \ell\} \rightarrow S \rightarrow S$ 
```

```
< s = s
```

```
_> :  $\forall \{\ell\} \{S : \mathbf{Set} \ell\} \rightarrow S \rightarrow S \times (1 \{\ell\})$ 
```

```
s > = s , tt
```

The following instances of grouping types demonstrate how information moves from the body level to the parameter level.

```
 $\mathcal{N}^0$  : DynamicSystem :waist 0
```

```
 $\mathcal{N}^0$  = <  $\mathbb{N}$  , 0 , suc >
```

```
 $\mathcal{N}^1$  : (DynamicSystem :waist 1)  $\mathbb{N}$ 
```

```
 $\mathcal{N}^1$  = < 0 , suc >
```

```

 $\mathcal{N}^2$  : (DynamicSystem :waist 2)  $\mathbb{N}$  0
 $\mathcal{N}^2$  = ⟨ suc ⟩

```

```

 $\mathcal{N}^3$  : (DynamicSystem :waist 3)  $\mathbb{N}$  0 suc
 $\mathcal{N}^3$  = ⟨ ⟩

```

Using `:waist i` we may fix the first i -parameters ahead of time. Indeed, the type `(DynamicSystem :waist 1) \mathbb{N}` is the type of dynamic systems over carrier \mathbb{N} , whereas `(DynamicSystem :waist 2) \mathbb{N} 0` is the type of dynamic systems over carrier \mathbb{N} and start state 0.

Examples of the need for such on-the-fly unbundling can be found in numerous places in the Haskell standard library. For instance, the standard libraries have two isomorphic copies of the integers, called `Sum` and `Prod`, whose reason for being is to distinguish two common monoids: The former is for *integers with addition* whereas the latter is for *integers with multiplication*. An orthogonal solution would be to use contexts:

```

Monoid : ∀  $\ell$  → Context ( $\ell$  suc  $\ell$ )
Monoid  $\ell$  = do Carrier ← Set  $\ell$ 
           Id      ← Carrier
           _⊕_     ← (Carrier → Carrier → Carrier)
           leftId  ← ∀ {x : Carrier} → x ⊕ Id ≡ x
           rightId ← ∀ {x : Carrier} → Id ⊕ x ≡ x
           assoc   ← ∀ {x y z} → (x ⊕ y) ⊕ z ≡ x ⊕ (y ⊕ z)
           End { $\ell$ }

```

With this context, `(Monoid ℓ_0 :waist 2) M ⊕` is the type of monoids over *particular* types M and *particular* operations \oplus . Of-course, this is orthogonal, since traditionally unification on the carrier type M is what makes typeclasses and canonical structures [9] useful for ad-hoc polymorphism.

4 TERMTYPES AS FIXED-POINTS

We have a practical monadic syntax for possibly parameterised record types that we would like to extend to `termtypes`. Algebraic data types are a means to declare concrete representations of the least fixed-point of a functor.

In particular, the description language D for dynamical systems, `??`, declares concrete constructors for the fixpoint of F :

```

F : Set → Set
F = λ (D : Set) → D ⊕ D

```

That is, $D \cong \text{Fix } F$ where:


```

data Fix (F : Set → Set) : Set where
  μ : F (Fix F) → Fix F

```

The problem is whether we can derive F from `DynamicSystem`. Let us attempt a quick calculation.

```

do X ← Set; z ← X; s ← (X → X); End
→{- Use existing interpretation to obtain a record. -}
  Σ X : Set • Σ z : X • Σ s : (X → X) • 1
→{- Pull out the carrier, “:waist 1”, to obtain a type constructor using “Π→λ”. -}
  λ X : Set • Σ z : X • Σ s : (X → X) • 1
→{- Termtypes constructors target the declared type, so only their sources matter.
    E.g., ‘z : X’ is a nullary constructor targeting the carrier ‘X’.
    This introduces 1 types, so any existing occurrences are dropped via 0.
-}
  λ X : Set • Σ z : 1 • Σ s : X • 0
→{- Termtypes are sums of products. -}
  λ X : Set • 1 ⊔ X ⊔ 0
→{- Termtypes are fixpoints of type constructors. -}
  Fix (λ X • 1 ⊔ X) -- i.e., D

```

Since we may view an algebraic data-type as a fixed-point of the functor obtained from the union of the sources of its constructors, it suffices to treat the fields of a record as constructors, then obtain their sources, then union them. That is, since algebraic-datatype constructors necessarily target the declared type, they are determined by their sources. For example, considered as a unary constructor $\text{op} : A \rightarrow B$ targets the type termtype B and so its source is A .

$\Downarrow \tau$ = “reduce all de brujin indices by 1”

$$\Sigma \rightarrow \uplus (\Sigma a : A \bullet Ba) = A \uplus \Sigma \rightarrow \uplus (\Downarrow Ba)$$

$$\text{sources } (\lambda x : (\Pi a : A \bullet Ba) \bullet \tau) = (\lambda x : A \bullet \text{sources } \tau)$$

$$\text{sources } (\lambda x : A \bullet \tau) = (\lambda x : 1 \bullet \text{sources } \tau)$$

```

{- Extend “sources, Σ→⊔” homomorphically to other syntactic constructs -}

```

```

termtype τ = Fix (Σ→⊔ (sources τ))

```

It is instructive to visually see how D is obtained from termtype in order to demonstrate that this approach to algebraic data types is practical.

```

D = termtype (DynamicSystem :waist 1)

-- Pattern synonyms for more compact presentation
pattern startD =  $\mu$  (inj1 tt)      -- : D
pattern nextD e =  $\mu$  (inj2 (inj1 e)) -- : D → D

```

With the pattern declarations, we can actually use these more meaningful names, when pattern matching, instead of the seemingly daunting μ -inj-ctions. For instance, we can immediately see that the natural numbers act as the description language for dynamical systems:

```

to : D → ℕ
to startD    = 0
to (nextD x) = suc (to x)

from : ℕ → D
from zero    = startD
from (suc n) = nextD (from n)

```

Astonishingly, useful programming datatypes arise from termtypes of theories (contexts). That is, if $C : \text{Set} \rightarrow \text{Context } \ell_0$ then $C' = \lambda X \rightarrow \text{termtype } (C X : \text{waist } 1)$ can be used to form ‘free, lawless, C -instances’.

Table 2. Data structures as free theories

Theory	Termtype
Dynamical Systems	\mathbb{N}
Pointed Structures	Maybe
Monoids	Binary Trees

The final entry in the table is a well known correspondence, that we can, not only formally express, but also prove to be true. We present the setup and leave it as an instructive exercise to the reader to present a bijective pair of functions between M and TreeSkeleton . Hint: Interactively case-split on values of M until the declared patterns appear, then associate them with the constructors of TreeSkeleton .

```

M : Set
M = termtype (Monoid  $\ell_0$  :waist 1)

-- Pattern synonyms for more compact presentation
pattern emptyM      =  $\mu$  (inj1 tt)      -- : M
pattern branchM l r =  $\mu$  (inj2 (inj1 (l , r , tt))) -- : M → M → M
pattern absurdM a   =  $\mu$  (inj2 (inj2 (inj2 (inj2 a)))) -- absurd values of 0

```

```

data TreeSkeleton : Set where
  empty  : TreeSkeleton
  branch : TreeSkeleton → TreeSkeleton → TreeSkeleton

```

To obtain trees over some ‘value type’ Ξ , one must start at the theory of “monoids containing a given set Ξ ”. Similarly, by starting at “theories of pointed sets over a given set Ξ ”, the resulting termtype is the Maybe type constructor —another instructive exercise to the reader: Show $P \cong \text{Maybe}$.

```

PointedOver : Set → Context (ℓsuc ℓ0)
PointedOver  $\Xi$  = do Carrier ← Set ℓ0
                  point   ← Carrier
                  embed    ← ( $\Xi$  → Carrier)
                  End

```

```

P : Set → Set
P  $X$  = termtype (PointedOver  $X$  :waist 1)

```

```

-- Pattern synonyms for more compact presentation
pattern nothingP =  $\mu$  (inj1 tt)      -- :  $P$ 
pattern justP e  =  $\mu$  (inj2 (inj1 e)) -- :  $P \rightarrow P$ 

```

5 RELATED WORKS

Surprisingly, conflating parameterised and non-parameterised record types with termtypes *within a language in a practical fashion* has not been done before.

The PackageFormer [1, 2] editor extension reads contexts—in nearly the same notation as ours— enclosed in dedicated comments, then generates and imports Agda code from them seamlessly in the background whenever typechecking transpires. The framework provides a fixed number of meta-primitives for producing arbitrary notions of grouping mechanisms, and allows arbitrary Emacs Lisp [6] to be invoked in the construction of complex grouping mechanisms.

The original PackageFormer paper provided the syntax necessary to form useful grouping mechanisms but was shy on the semantics of such constructs. We have chosen the names of our combinators to closely match those of PackageFormer’s with an aim of furnishing the mechanism with semantics by construing the syntax as semantics-functions; i.e., we have a shallow embedding of PackageFormer’s constructs as Agda entities:

PackageFormer’s `_:kind_` meta-primitive dictates how an abstract grouping mechanism should be viewed in terms of existing Agda syntax. However, unlike PackageFormer, all of our syntax consists of legitimate Agda terms. Since language syntax is being manipulated, we are forced to define it as a macro:

Table 3. Comparing the in-language Context mechanism with the PackageFormer editor extension

	PackageFormer	Contexts
Type of Entity	Preprocessing Tool	Language Library
Specification Language	Lisp + Agda	Agda
Well-formedness Checking	✗	✓
Termination Checking	✓	✓
Elaboration Tooltips	✓	✗
Rapid Prototyping	✓	✓ (Slower)
Usability Barrier	None	None
Extensibility Barrier	Lisp	Weak Metaprogramming

Table 4. Contexts as a semantics for PackageFormer constructs

Syntax	Semantics
PackageFormer	Context
:waist	:waist
\rightarrow	Forward function application
:kind	:kind, see below
:level	Agda built-in
:alter-elements	Agda macros

```
data Kind : Set where
```

```
  'record   : Kind
```

```
  'typeclass : Kind
```

```
  'data     : Kind
```

```
C :kind 'record   = C 0
```

```
C :kind 'typeclass = C :waist 1
```

```
C :kind 'data     = termtype (C :waist 1)
```

We did not expect to be able to assign a full semantics to PackageFormer’s syntactic constructs due to Agda’s substantially weak metaprogramming mechanism. However, it is important to note that PackageFormer’s Lisp extensibility expedites the process of trying out arbitrary grouping mechanisms—such as partial-choices of pushouts and pullbacks along user-provided assignment functions—since it is all either string or symbolic list manipulation. On the Agda side, using contexts, it would require exponentially more effort due to the limited reflection mechanism and the intrusion of the stringent type system.

6 CONCLUSION

Starting from the insight that related grouping mechanisms could be unified, we showed how related structures can be obtained from a single declaration using a practical interface. The resulting framework, based on contexts, still captures the familiar record declaration syntax as well as the expressivity of usual

algebraic datatype declarations—at the minimal cost of using pattern declarations to aide as user-chosen constructor names. We believe that our approach to using contexts as general grouping mechanisms *with* a practical interface are interesting contributions.

We used the focus on practicality to guide the design of our context interface, and provided interpretations both for the rather intuitive “contexts are name-type records” view, and for the novel “contexts are fixed-points” view for termtypes. In addition, to obtain parameterised variants, we needed to explicitly form “contexts whose contents are over a given ambient context”—e.g., contexts of vector spaces are usually discussed with the understanding that there is a context of fields that can be referenced— which we did using monads.

To those interested in exotic ways to group data together—such as, mechanically deriving product types and homomorphism types of theories— we offer an interface that is extensible using Agda’s reflection mechanism. In comparison with, for example, special-purpose preprocessing tools, this has obvious advantages in accessibility and semantics.

To Agda programmers, this offers a standard interface for grouping mechanisms that had been sorely missing, with an interface that is so familiar that there would be little barrier to its use. In particular, as we have shown, it acts as an in-language library for exploring relationships between free theories and data structures. As we have only presented the high-level definitions of the core combinators, leaving the Agda-specific details to the appendices, it is also straightforward to translate the library into other dependently-typed languages.

7 APPENDIX: WHAT ABOUT THE META-LANGUAGE’S PARAMETERS? MAYBE_DELETE

Besides `:waist`, another way to introduce parameters into a context grouping mechanism is to use the language’s existing utility of parameterising a context by another type—as was done earlier in `PointedOver`.

For example, a pointed set needn’t necessarily be terminated with `End`.

```
PointedSet : Context  $\ell_1$ 
PointedSet = do Carrier  $\leftarrow$  Set
              point  $\leftarrow$  Carrier
              End { $\ell_1$ }
```

We instead form a grouping consisting of a single type and a value of that type, along with an instance of the parameter type Ξ .

```
PointedPF : ( $\Xi$  : Set1)  $\rightarrow$  Context  $\ell_1$ 
PointedPF  $\Xi$  = do Carrier  $\leftarrow$  Set
                point  $\leftarrow$  Carrier
                ‘  $\Xi$ 
```

Clearly `PointedPF 1 \approx PointedSet`, so we have a more generic grouping mechanism. The natural next step is to consider other parameters such as `PointedSet` in-place of Ξ .

```

-- Convenience names
PointedSetr = PointedSet           :kind 'record
PointedPFr = λ Ξ → PointedPF Ξ :kind 'record

-- An extended record type: Two types with a point of each.
TwoPointedSets = PointedPFr PointedSetr

_ : TwoPointedSets
≡ ( Σ Carrier1 : Set • Σ point1 : Carrier1
    • Σ Carrier2 : Set • Σ point2 : Carrier2 • 1)
_ = refl

-- Here's an instance
one : PointedSet :kind 'record
one = ℬ , false , tt

-- Another; a pointed natural extended by a pointed bool,
-- with particular choices for both.
two : TwoPointedSets
two = ℕ , 0 , one

```

More generally, *record **structure** can be dependent on values*:

```

_PointedSets : ℕ → Set1
zero PointedSets = 1
suc n PointedSets = PointedPFr (n PointedSets)

_ : 4 PointedSets
≡ (Σ Carrier1 : Set • Σ point1 : Carrier1
    • Σ Carrier2 : Set • Σ point2 : Carrier2
    • Σ Carrier3 : Set • Σ point3 : Carrier3
    • Σ Carrier4 : Set • Σ point4 : Carrier4 • 1)
_ = refl

```

Using traditional grouping mechanisms, it is difficult to create the family of types n PointedSets since the number of fields, $2 \times n$, depends on n .

It is interesting to note that the termtype of PointedPF is the same as the termtype of PointedOver, the Maybe type constructor!

```
PointedD : (X : Set) → Set1
PointedD X = termtype (PointedPF (Lift _ X) :waist 1)
```

```
-- Pattern synonyms for more compact presentation
```

```
pattern nothingP = μ (inj1 tt)
pattern justP x   = μ (inj2 (lift x))
```

```
casingP : ∀ {X} (e : PointedD X)
  → (e ≡ nothingP) ⊔ (Σ x : X • e ≡ justP x)
casingP nothingP = inj1 refl
casingP (justP x) = inj2 (x , refl)
```

REFERENCES

- [1] Musa Al-hassy. The next 700 module systems: Extending dependently-typed languages to implement module system features in the core language, 2019. URL <https://alhassey.github.io/next-700-module-systems-proposal/thesis-proposal.pdf>.
- [2] Musa Al-hassy, Jacques Carette, and Wolfram Kahl. A language feature to unbundle data at will (short paper). In Ina Schaefer, Christoph Reichenbach, and Tijs van der Storm, editors, *Proceedings of the 18th ACM SIGPLAN International Conference on Generative Programming: Concepts and Experiences, GPCE 2019, Athens, Greece, October 21-22, 2019*, pages 14–19. ACM, 2019. ISBN 978-1-4503-6980-0. doi: 10.1145/3357765.3359523. URL <https://doi.org/10.1145/3357765.3359523>.
- [3] Richard Bird. Thinking functionally with haskell. 2009. doi: 10.1017/cbo9781316092415. URL <http://dx.doi.org/10.1017/cbo9781316092415>.
- [4] Ana Bove, Peter Dybjer, and Ulf Norell. A brief overview of Agda — A functional language with dependent types. In *Theorem Proving in Higher Order Logics, 22nd International Conference, TPHOLs 2009, Munich, Germany, August 17–20, 2009. Proceedings*, pages 73–78, 2009. doi: 10.1007/978-3-642-03359-9_6.
- [5] François Garillot, Georges Gonthier, Assia Mahboubi, and Laurence Rideau. Packaging Mathematical Structures. In Tobias Nipkow and Christian Urban, editors, *Theorem Proving in Higher Order Logics*, volume 5674 of *Lecture Notes in Computer Science*, Munich, Germany, 2009. Springer. URL <https://hal.inria.fr/inria-00368403>.
- [6] Paul Graham. *ANSI Common Lisp*. Prentice Hall Press, USA, 1995. ISBN 0133708756.
- [7] Jason Gross, Adam Chlipala, and David I. Spivak. Experience Implementing a Performant Category-Theory Library in Coq, 2014.
- [8] Paul Hudak, John Hughes, Simon L. Peyton Jones, and Philip Wadler. A history of haskell: being lazy with class. In Barbara G. Ryder and Brent Hailpern, editors, *Proceedings of the Third ACM SIGPLAN History of Programming Languages Conference (HOPL-III), San Diego, California, USA, 9-10 June 2007*, pages 1–55. ACM, 2007. doi: 10.1145/1238844.1238856. URL <https://doi.org/10.1145/1238844.1238856>.
- [9] Assia Mahboubi and Enrico Tassi. Canonical Structures for the working Coq user. In Sandrine Blazy, Christine Paulin, and David Pichardie, editors, *ITP 2013, 4th Conference on Interactive Theorem Proving*, volume 7998 of *LNCS*, pages 19–34, Rennes, France, July 2013. Springer. doi: 10.1007/978-3-642-39634-2_5. URL <https://hal.inria.fr/hal-00816703>.
- [10] Eugenio Moggi. Notions of computation and monads. *Inf. Comput.*, 93(1):55–92, 1991. doi: 10.1016/0890-5401(91)90052-4. URL [https://doi.org/10.1016/0890-5401\(91\)90052-4](https://doi.org/10.1016/0890-5401(91)90052-4).
- [11] Ulf Norell. *Towards a Practical Programming Language Based on Dependent Type Theory*. PhD thesis, Dept. Comp. Sci. and Eng., Chalmers Univ. of Technology, September 2007.